

The Mass-Spring System

Warren Weckesser

Department of Mathematics, Colgate University

This Maple session uses the mass-spring system to demonstrate the phase plane, direction fields, solution curves (“trajectories”), and the extended phase space.

The differential equation is

$$m \left(\frac{\partial^2}{\partial t^2} y(t) \right) + b \left(\frac{\partial}{\partial t} y(t) \right) + k y(t) = 0,$$

where $y(t)$ is the displacement of the mass from its equilibrium position, m is the mass of object attached to the spring, b is the coefficient of friction (or damping), and k is the spring constant.

We convert this to a system by defining

$$v(t) = \frac{\partial}{\partial t} y(t)$$

Then

$$\frac{\partial}{\partial t} v(t) = -\frac{k y(t)}{m} - \frac{b \left(\frac{\partial}{\partial t} y(t) \right)}{m}$$

or

$$\frac{\partial}{\partial t} v(t) = -\frac{k y(t)}{m} - \frac{b v(t)}{m}$$

This equation along with the equation that defines $v(t)$ make up a system of two first order differential equations.

```
[ > with(DEtools):
```

```
[ We first define the two equations that make up the system:
```

```
[ > eq1 := diff(y(t),t) = v(t);
```

$$eq1 := \frac{\partial}{\partial t} y(t) = v(t)$$

```
[ > eq2 := diff(v(t),t) = -(k/m)*y(t) - (b/m)*v(t);
```

$$eq2 := \frac{\partial}{\partial t} v(t) = -\frac{k y(t)}{m} - \frac{b v(t)}{m}$$

```
[ Now define a variable that holds the system:
```

```
[ > sys := [eq1,eq2];
```

$$\text{sys} := \left[\frac{\partial}{\partial t} y(t) = v(t), \frac{\partial}{\partial t} v(t) = -\frac{k y(t)}{m} - \frac{b v(t)}{m} \right]$$

We substitute in some numerical values for the parameters, so we can plot some solutions.

The "subs" command makes the substitutions given by the first several arguments in the expression given as the last argument, so the following command replacing substitute $b=1/3$, $m=1$ and $k=1$ into sys, and calls the result sys1.


```
> sys1 := subs(b=1/3,m=1,k=1,sys);
```

$$\text{sys1} := \left[\frac{\partial}{\partial t} y(t) = v(t), \frac{\partial}{\partial t} v(t) = -y(t) - \frac{1}{3} v(t) \right]$$

Let's first look at the direction field. This is the same as the vector field, but all the vectors are plotted with the same length. We use the DEplot command.

```
> DEplot(sys1,[y(t),v(t)],t=0..1,y=-1..1,v=-1..1,scaling=constrained,
,color=black,title="Mass-Spring Direction Field (m=1,k=1,b=1/3)");
```

Mass-Spring Direction Field (m=1,k=1,b=1/3)




Let's add a solution curve, also called a *trajectory*. We will find the curve with initial condition $y(0)=1$, $v(0)=0$. The ics argument to DEplot is a list of triples of the form $[t_0, y_0, v_0]$. Here we only put in one initial condition, $[0,1,0]$.

```
> ics := [[0,1,0]];
```

```
ics := [[0, 1, 0]]
```


```
> DEplot(sys1,[y(t),v(t)],t=-5..15,ics,y=-1..1,v=-1..1,scaling=constrained,
,color=black,title="Mass-Spring Direction Field (m=1,k=1,b=1/3)",linecolor=blue,stepsize=.1);
```

Mass-Spring Direction Field ($m=1, k=1, b=1/3$)



Let's plot $y(t)$ vs. t for the same solution. The option "scene=[var1,var2]" let's us choose how the solution is plotted. If we say "scene=[t,y]", we get a plot of $y(t)$ vs. t .


```
> DEplot(sys1,[y(t),v(t)],t=-5..15,ics,y=-2.5..1.5,v=-1..1,color=black,
title="Mass-Spring
(m=1,k=1,b=1/3)",linecolor=blue,stepsize=.1,scene=[t,y]);
```



Now let's see this solution in the extended phase space. Here we use the command **DEplot3d**. (Don't forget that you can click-and-drag in the 3D plot to see different views.) Note that, as above, we are plotting the solution for $-5 < t < 15$. The solution is defined in forward and backward time.

```
> DEplot3d(sys1,[y(t),v(t)],t=-5..15,ics,y=-2.5..1.5,v=-1..1,color=black,
title="Mass-Spring Extended Phase Space
(m=1,k=1,b=1/3)",linecolor=blue,stepsize=.1,scene=[t,y,v],axes=normal);
```

Mass-Spring Extended Phase Space ($m=1, k=1, b=1/3$)



By choosing appropriate viewing angle of the 3D plot, you can see that the previous two plots (the phase plane and the plot of $y(t)$ vs. t) are just different projections of the solution curve in the extended phase space.


Let's add another curve, with initial conditions $y(0)=-0.5$, $v(0)=0.5$. Also, we'll only plot $0 < t < 15$.

```
> ics := [[0,1,0],[0,-0.5,0.5]];
```

```
ics := [[0,1,0],[0,-.5,.5]]
```

```
> DEplot3d(sys1,[y(t),v(t)],t=0..15,ics,y=-2.5..1.5,v=-1..1,color=black,
title="Mass-Spring Extended Phase Space
(m=1,k=1,b=1/3)",linecolor=blue,stepsize=.1,scene=[t,y,v],axes=normal);
```

Mass-Spring Extended Phase Space ($m=1, k=1, b=1/3$)



One more plot. This time we'll plot two solutions that have the same initial values, but the starting time is different.

One solution satisfies $y(0)=1$, $v(0)=0$, and the other satisfies $y(2)=1$, $v(2)=0$.


```
> ics := [[0,1,0],[2,1,0]];
```

```

[
[
    ics := [[0, 1, 0], [2, 1, 0]]
> DEplot3d(sys1,[y(t),v(t)],t=0..15,ics,y=-2.5..1.5,v=-1..1,color=black,
title="Mass-Spring Extended Phase Space
(m=1,k=1,b=1/3)",linecolor=blue,stepsize=.1,scene=[t,y,v],axes=normal);

```

Mass-Spring Extended Phase Space (m=1,k=1,b=1/3)



When you view these solutions from the phase plane perspective, you see only one curve. These two solutions follow the same trajectory, but at different times.

>